

```
* Stata Analysis Examples Replication for ASDA 3rd Edition
* Berglund Fall 2024
```

```
*** CHAPTER 10
```

```
use "P:\ASDA3\Data Sets for Analysis Examples and Stata R Code\ncsr.dta", clear
```

```
* K-M Analysis
```

```
gen ageonsetmde = age
replace ageonsetmde = MDE_OND if mde == 1
```

```
stset ageonsetmde [pweight = ncsrwtsh], failure(mde==1)
```

```
* Figure 10.3
```

```
sts graph, by(racecat) legend(lab(1 "Other/Asian") ///
lab(2 "Hispanic/Mexican") lab(3 "Black") lab(4 "White"))
```

```
sts list, at(10 20 30 40 50 60 70)
```

```
* Example 10.4.5 Cox PH Model
```

```
svyset seclustr [pweight=ncsrwtsh], strata(sestrat)
stset ageonsetmde [pweight = ncsrwtsh], failure(mde==1)
```

```
svy: stcox age ib2.sex i.MAR3CAT i.ED4CAT i.racecat
```

```
* Figure 10.4
```

```
stphplot, by(racecat) legend(lab(1 "Other/Asian") ///
lab(2 "Hispanic/Mexican") lab(3 "Black") lab(4 "White"))
```

```
* Figure 10.5
```

```
gen prevmarr = .
replace prevmarr = 1 if MAR3CAT == 2
replace prevmarr = 0 if MAR3CAT == 1 | MAR3CAT == 3
gen nevmarr = .
replace nevmarr = 1 if MAR3CAT == 3
replace nevmarr = 0 if MAR3CAT == 1 | MAR3CAT == 2
gen ed12 = .
replace ed12 = 1 if ED4CAT == 2
replace ed12 = 0 if ED4CAT == 1 | ED4CAT == 3 | ED4CAT == 4
gen ed1315 = .
replace ed1315 = 1 if ED4CAT == 3
replace ed1315 = 0 if ED4CAT == 1 | ED4CAT == 2 | ED4CAT == 4
gen ed16 = .
replace ed16 = 1 if ED4CAT == 4
replace ed16 = 0 if ED4CAT == 1 | ED4CAT == 2 | ED4CAT == 3
```

```
stphplot, by(racecat) adj(age sexf prevmarr nevmarr ///
ed12 ed1315 ed16) nolntime ///
legend(lab(1 "Other/Asian") lab(2 "Hispanic/Mexican") ///
lab(3 "Black") lab(4 "White"))
```

```
* Discrete time models
```

```
expand age
sort caseid
by caseid: gen pyr = _n
```

```
gen mdetv = 1 if pyr == MDE_OND
replace mdetv = 0 if pyr != MDE_OND
```

```
list caseid age ncsrwtsh sestrat seclustr pyr mdetv ///
ageonsetmde if caseid == 1
```

```
* Table 10.5
```

```
svy: logistic mdetv pyr age ib2.sex i.ED4CAT ///
i.racecat i.MAR3CAT if pyr <= ageonsetmde
```

```
* Table 10.6
```

```
svy: cloglog mdetv pyr age ib2.sex i.ED4CAT ///
i.racecat i.MAR3CAT if pyr <= ageonsetmde, eform
```